

**TI-82  
TI-83  
TI-83 Plus**

## Systems of Linear Equations Program

This program will display the solution of a system of two linear equations in two variables of the form

$$ax + by = c$$

$$dx + ey = f$$

if a unique solution exists.

```
PROGRAM:SOLVE
:Disp "AX+BY=C"
:Prompt A
:Prompt B
:Prompt C
:Disp "DX+EY=F"
:Prompt D
:Prompt E
:Prompt F
:If AE-DB=0
:Then
:Disp "NO UNIQUE"
:Disp "SOLUTION"
:Else
:(CE-BF)/(AE-DB)→X
:(AF-CD)/(AE-DB)→Y
:Disp X
:Disp Y
:End
```

## Adding Vectors Graphically Program

This program will sketch two vectors in standard position. Using the parallelogram law for the vector addition, the program also sketches the vector sum. Be sure to set an appropriate viewing rectangle.

```
PROGRAM:ADDVECT
:ClrDraw
:Input "ENTER A",A
:Input "ENTER B",B
:Input "ENTER C",C
:Input "ENTER D",D
:Line(0,0,A,B)
:Line(0,0,C,D)
:A+C→E
:B+D→F
:Line(0,0,E,F)
:Line(A,B,E,F)
:Line(C,D,E,F)
:Pause
:Stop
```

## Visualizing Row Operations Program

This program demonstrates how elementary matrix row operations used in Gauss-Jordan elimination may be interpreted graphically. It asks the user to enter a  $2 \times 3$  matrix that corresponds to a system of two linear equations. (The matrix entries should not be equivalent to either vertical or horizontal lines. This demonstration is also most effective if the  $y$ -intercepts of the lines are between  $-10$  and  $10$ .)

While the demonstration is running, you should notice that each elementary row operation creates an equivalent system. This equivalence is reinforced graphically because, although the equations of the lines change with each elementary row operation, the point of intersection remains the same. You may want to run this program a second time to notice the relationship between the row operations and the graphs of the lines of the system. To use this program, enter the order of matrix [A] as a  $2 \times 3$  matrix. Press ENTER after each screen display to continue the program.

```

PROGRAM: ROWOPS
:Disp "ENTER A"
:Disp "2 BY 3 MATRIX:"
:Disp "A B C"
:Disp "D E F"
:Prompt A,B,C
:Prompt D,E,F
:A→[A](1,1):B→[A](1,2)
:C→[A](1,3):D→[A](2,1)
:E→[A](2,2):F→[A](2,3)
:ClrHome
:Disp "ORIGINAL MATRIX:"
:Pause [A]
:“(B-1(C-AX))”→Y2
:“(E-1(F-DX))”→Y1
:ZStandard:Pause:ClrHome
:Disp "OBTAIN LEADING"
:Disp "1 IN ROW 1"
:*row(A-1,[A],1)→[A]
:Pause [A]:ClrDraw
:“(A/B)(C/A-X)”→Y2
:DispGraph:Pause:ClrHome
:Disp "OBTAIN 0 BELOW"
:Disp "LEADING 1 IN"
:Disp "COLUMN 1"
:*row+(-D,[A],1,2)→[A]
:Pause [A]:ClrDraw
:“(E-(BD/A))-1(F-(DC/A))”→Y1
:DispGraph:Pause:ClrHome

:[A](2,2)→G
:If G=0
:Goto 1
:*row(G-1,[A],2)→[A]
:Disp "OBTAIN LEADING"
:Disp "1 IN ROW 2"
:Pause [A]:ClrDraw
:DispGraph:Pause:ClrHome
:Disp "OBTAIN 0 ABOVE"
:Disp "LEADING 1 IN"
:Disp "COLUMN 2"
:[A](1,2)→H
:*row+(-H,[A],2,1)→[A]
:Pause [A]:ClrDraw:FnOff 2
:Vertical -(B/A)(E-(BD/A))-1(F-DC/A)+C/A
:DispGraph:Pause:ClrHome
:Disp "THE POINT OF"
:Disp "INTERSECTION IS"
:Disp "X=", [A](1,3), "Y=", [A](2,3)
:Stop
:Lbl 1
:If [A](2,3)=0
:Then
:Disp "INFINITELY MANY"
:Disp "SOLUTIONS"
:Else
:Disp "INCONSISTENT"
:Disp "SYSTEM"
:End

```

**TI-89  
TI-92  
TI-92 Plus  
Voyage 200**

## Systems of Linear Equations Program

This program will display the solution of a system of two linear equations in two variables of the form

$$ax + by = c$$

$$dx + ey = f$$

if a unique solution exists.

```
:solvelin( )
:Prgm
:ClrIO
:Disp "Ax+By=C"
:Input "Enter A.",a
:Input "Enter B.",b
:Input "Enter C.",c
:ClrIO
:Disp "Dx+Ey=F"
:Input "Enter D.",d
:Input "Enter E.",e
:Input "Enter F.",f
:If a*e-d*b=0 Then
:  Disp "No unique solution"
: Else
:  (c*e-b*f)/(a*e-d*b)→x
:  (a*f-c*d)/(a*e-d*b)→y
:  Disp x
:  Disp y
:EndIf
:EndPrgm
```

## Adding Vectors Graphically Program

This program will sketch two vectors in standard position. Using the parallelogram law for the vector addition, the program also sketches the vector sum. Be sure to set an appropriate viewing rectangle.

```
:addvect( )
:Prgm
:ClrIO
:Input "ENTER a",a
:Input "ENTER b",b
:Input "ENTER c",c
:Input "ENTER d",d
:ClrDraw
:a+c→e
:b+d→f
:Line 0,0,a,b
:Line 0,0,c,d
:Line 0,0,e,f
:Line a,b,e,f
:Line c,d,e,f
:Pause
:setMode("Split 1 App","Home")
:Stop
:EndPrgm
```

## Visualizing Row Operations Program

This program demonstrates how elementary matrix row operations used in Gauss-Jordan elimination may be interpreted graphically. It asks the user to enter a  $2 \times 3$  matrix that corresponds to a system of two linear equations. (The matrix entries should not be equivalent to either vertical or horizontal lines. This demonstration is also most effective if the  $y$ -intercepts of the lines are between  $-10$  and  $10$ .)

While the demonstration is running, you should notice that each elementary row operation creates an equivalent system. This equivalence is reinforced graphically because, although the equations of the lines change with each elementary row operation, the point of intersection remains the same. You may want to run this program a second time to notice the relationship between the row operations and the graphs of the lines of the system. Press  after each screen display to continue the program.

```
:rowops( )
:Prgm
:ClrIO
:ClrHome
:setMode("Split Screen","Left-Right")
:setMode("Split 1 App","Home")
:setMode("Split 2 App","Graph")
:Disp "ENTER A"
:Disp "2 BY 3 MATRIX:"
:Disp "A B C"
:Disp "D E F"
:Prompt a,b,c
:Prompt d,e,f
:[[a,b,c][d,e,f]]→mat1
:ClrIO
:b^(-1)*(c-a*x)→y2(x)
:e^(-1)*(f-d*x)→y1(x)
:ZoomStd
:Disp "ORIGINAL MATRIX:"
:Pause mat1
:ClrIO
:a/b*(c/a-x)→y2(x)
:Disp "OBTAIN LEADING"
:Disp "1 IN ROW 1"
:mRow(a^(-1),mat1,1)→mat1
:Pause mat1
:ClrIO
:(e-b*d/a)^(-1)*(f-d*c/a)→y1(x)
:DispG
:Disp "OBTAIN 0 BELOW"
:Disp "LEADING 1 IN"
:Disp "COLUMN 1"
:mRowAdd(-d,mat1,1,2)→mat1
:Pause mat1

:ClrIO
:mat1[2,2]→g
:If g=0
:Goto a1
:Disp "OBTAIN LEADING"
:Disp "1 IN ROW 2"
:mRow(g^(-1),mat1,2)→mat1
:Pause mat1
:ClrIO
:mat1[1,2]→h
:FnOff 2
:LineVert -b/a*(e-b*d/a)^(-1)*(f-d*c/a)+c/a
:Disp "OBTAIN 0 ABOVE"
:Disp "LEADING 1 IN"
:Disp "COLUMN 2"
:mRowAdd(-h,mat1,2,1)→mat1
:Pause mat1
:ClrIO
:Disp "THE POINT OF"
:Disp "INTERSECTION IS"
:Disp "X=",mat1[1,3],"Y=",mat1[2,3]
:Goto A2
:Lbl a1
:If mat1[2,3]=0 Then
:Disp "INFINITELY MANY"
:Disp "SOLUTIONS"
:Else
:Disp "INCONSISTENT"
:Disp "SYSTEM"
:EndIf
:Lbl A2
:Pause
:setMode("Split Screen","Full")
:EndPrgm
```